

Commentaires du code

- On définit l'espace de travail

1: UFRAME_NUM=2 ; / Utilisation de l'userframe numéro 2

2: UTOOL_NUM=2 ; / Utilisation de l'usertool numéro 2 -> définition de notre environnement de travail

- Définition des variables globales

3: PR[112]=PR[117] ; / Nous définissons une variable de travail (112). On lui assigne la valeur de la variable 117 que nous ne devons surtout pas écraser !

4: PR[111]=PR[115] ; / Idem avec 111 et 115. Le PR[111] définit notre rangée de travail (A->D / E->G / H->I). On commence ici avec la rangée A->D.

5: J PR[110] 50% CNT0 ; / Pour travailler nous avons besoin d'un point de départ (ici notre PR[110]) qui est une mesure de sécurité. Nous avons choisi une CTN0 car nous voulons qu'il passe réellement sur le point, mais sans marquer d'arrêt complet.

- Conditions de déplacements

6: FOR R[160]=0 TO 8 ; / Notre R[160] est notre index de boucle for. Il s'incrémentera de 0 à 8 soit de A à I.

7: IF R[160]<>4,JMP LBL[5] ; / Si nous travaillons sur la pièce E on reste sur la rangée A->D.

8: PR[111]=PR[120] ; / Sinon, on change la valeur de notre variable de travail pour passer à la rangée supérieure (E->G).

9: LBL[5] ;

10: IF R[160]<>7,JMP LBL[6] ; / Si nous travaillons sur la pièce H on reste sur la rangée E->G.

11: PR[111]=PR[121] ; / Sinon, on change la valeur de notre variable de travail pour passer à la rangée supérieure (H->I).

12: LBL[6] ;

13: IF R[160]=0,JMP LBL[7] ; / Si nous travaillons sur la pièce A on reste sur la rangée H->I.

14: PR[111]=PR[111]+PR[114] ; / Ici on attribue la position de notre variable de travail pour passer de lettre en lettre sur une même rangée (par exemple de A->B) (nous restons au-dessus de ces dernières par sécurité)

15: LBL[7] ;

- Déplacement et saisi de la pièce

16: J PR[111] 50% CNT0 ; / On se rend sur notre variable de travail (c'est à dire au-dessus d'une lettre)

17: J PR[111] 50% CNT0 Offset, PR[113] ; / On descend sur la pièce

18: CALL HAND_CLOSE ; / On appelle le programme de fermeture de pince pour attraper la pièce

19:J PR[111] 50% CNT20 ; / **On remonte avec la pièce "en main".**

- Empilement les pièces

20: IF R[160]=0,JMP LBL[1] ; / **Si nous bougeons la pièce A, on commencera la pile au sol**

21: PR[112]=PR[112]+PR[116] ; / **Sinon on attribue la position de notre variable de travail pour modifier la hauteur de la pile selon la pièce agrippée.**

22: LBL[1] ;

23:J PR[112] 50% CNT20 ; / **On se déplace vers le point M (lieu de formation de la pile) à une hauteur adaptée à la pièce en déplacement**

24:J PR[112] 50% CNT0 Offset,PR[113] ; / **On pose la pièce délicatement**

25: CALL HAND_OPEN ; / **On appelle le programme d'ouverture de la pince afin de lâcher notre pièce**

26:J PR[112] 50% CNT20 ; / **On remonte pour ne pas tout faire tomber en repartant chercher la pièce suivante (mesure de sécurité)**

27: ENDFOR ;

Registers and position registers

R[160] = 0 : notre index pour la boucle for

| | | | |
|--|---------------------------|---|---------------------------|
| PR[110] : au-dessus du point F | | PR[111] : variable de travail elle sera donc écrasée dès la ligne 4 | |
| x = 0 y = -332 z = 100 | w = 180 p = 0 r = 0 | x = 79.5 y = -286 z = 40 | w = 180 p = 0 r = 0 |
| PR[112] : variable de travail elle sera donc écrasée dès la ligne 3 | | PR[113] : variable de variation de hauteur (pour descendre sur une pièce) | |
| x = 0 y = -92 z = 40 | w = 180 p = 0 r = 0 | x = 0 y = 0 z = -40 | w = 0 p = 0 r = 0 |
| PR[114] : variable de décalage entre chaque pièce d'une même rangée | | PR[115] : position au-dessus du point A | |
| x = -53 y = 0 z = 0 | w = 0 p = 0 r = 0 | x = 79.5 y = -286 z = 40 | w = 180 p = 0 r = 0 |
| PR[116] : variable de variation de hauteur entre chaque pièce (sens montant) | | PR[117] : position au-dessus du point M | |
| x = 0 y = 0 z = 40 | w = 0 p = 0 r = 0 | x = 0 y = -92 z = 40 | w = 180 p = 0 r = 0 |
| PR[120] : position au-dessus du point E | | PR[121] : position au-dessus du point H | |
| x = 106 y = -332 z = 40 | w = 180 p = 0 r = 0 | x = 79.5 y = -378 z = 40 | w = 180 p = 0 r = 0 |